Glass Audio 10th Anniversary Glass Audio

$6.00
CANADA $8.57

GLASS AUDIO

CONTENTS

Volume 10 *Number 2*
1998

40W TRIODE AND 60W ULTRALINEAR AMPLIFIER

BY JOSEPH NORWOOD STILL



PHOTO 1: The triode/UL amplifier.

The original version of this article appeared in Electronics World (USA), *July 1959, p. 60.*

From the resurgence of commercial hi-fi amplifiers employing tubes, it is obvious that the vacuum tube has been resurrected. Many of the new designs are in the moderate to very-expensive price range, but my amplifier (*Photo 1*) should satisfy the hi-fi requirements of all but the ultrapurist. If you're experienced, you can build it with a reasonable amount of work for about $500. The amplifier operates in either a triode 40W or an ultralinear (UL) 60W mode. For those enthusiasts who desire a great-sounding amplifier, here is one that can truly deliver.

Advantages of Triodes

Triodes offer major advantages in three realms: uniform response, damping characteristics, and distortion. First of all, an

to page 8

ALGEBRAIC TECHNIQUE FOR MODELING TRIODES

BY JEAN-CHARLES MAILLET

The idea of generating SPICE simulation models for vacuum-tube (VT) circuits has a lot of appeal for researchers who understand the limitations of the highly approximate classical theory. Unfortunately, most of the techniques currently used to model vacuum-tube devices fail to comply in several basic ways with modern practices of general device modeling.

First, these techniques are generally based on theoretical considerations instead of properly measured and averaged device-transfer data; second, the models address in an active way only the output-port current function, while the input port is generally passive and therefore functionally incorrect; third, the models are fixed in the sense that you cannot readily alter them for a customized match.

Early-Model Assumptions

Early VT SPICE models[1,2] were firmly based on assumptions made in tube-circuit theory that were primarily intended to provide analytically tractable equivalent-device replacements for use in hand analysis. These assumptions included a 3/2-power-based expression for plate current, and a zero-value expression for grid current.

The 3/2-power law derives from a one-dimensional representation of the tube's conduction channel, and does not mimic mu-factor dependence on terminal voltages. On the other hand, the idealized zero-grid-current assumption represents the view of a perfect or near-perfect voltage-sensing circuit. Both situations contribute to device-model behavior that cannot provide good estimates in large-signal situations—that is, in transient or DC-sweep simulations. More recent modeling efforts by Koren and Rydel[3,4] either introduce variant forms of the plate-current expression or build on the simpler form to derive a more accurate match to a supposed reference.

In order to generate device models useful for accurate transient, swept-DC, and AC simulations, not only must absolute current levels match, but they should also be close in the first-order (slope function or first derivative) of the transfer curves over the intended range of application and simulation. The divergence between the curvature of transfer curves of existing models and that of real devices was discussed by Pritchard,[5] but *Fig. 1* shows precisely how Koren's 12AX7 model performs against corresponding RCA transfer data. The fit is reasonable in the sense that the curves line up fairly well in terms of input-voltage values, although the curvature profile shows considerable divergence.



FIGURE 1: Koren 12AX7 SPICE model plate-transfer curves against RCA.

Another researcher has also published comparative results based on Koren's model without intending to display discrepancies, though the results strongly suggest otherwise.[6] The reason this matter is overlooked lies in the common tendency to overuse AC (Bode) analysis in general simulation work. This is underlined by a lack of familiarity with the linearization process that occurs in Laplacian analysis, the central mathematical principle used in this type of analysis.

In AC analysis, signal amplitudes are reduced to zero by a limiting process used to linearize all control-source transfer functions in the circuit. Hence, AC analysis provides a reduced view of what occurs in a circuit, and therefore has limited applications. Accurate transient (time-domain) analysis requires accurate wide-range modeling, since the dynamic characteristics of components and circuit-intercoupling mechanisms are exercised just as in a real circuit. Models that provide accurate transient analyses will also provide accurate AC analysis results, while the converse does not in general hold true.



FIGURE 2: General two-port network representation.



FIGURE 3: Terminal current functions corresponding to the two-port network formulation of a vacuum tube triode device.

Of prime importance to the issue of accurate modeling are the plate-resistance (R_p) levels in Koren's plate-circuit model corresponding to the first derivative of the I_p (V_{pc}) curves. Specifically, it does not match well with RCA's

TABLE 1

12AX7 PLATE-CIRCUIT CURRENT-TRANSFER DATA

$V_{gc} = +1.0$
0.0, 0.121 5, 1.60 18, 2.03 41, 2.66 79, 3.74 116, 4.79

$V_{gc} = +0.5$
0.0, 0.06 20, 1.04 59, 2.08 81, 2.53 118, 3.57 134, 4.0

$V_{gc} = 0.0$
0.0, 0.0 7, 0.28 17, 0.56 43, 1.07 68, 1.46 86, 1.79 119, 2.45 134, 2.79 158, 3.39 181, 3.98

$V_{gc} = -0.5$
0.0, 0.0 24, 0.18 59, 0.56 95, 1.08 122, 1.54 147, 2.02 173, 2.60 193, 3.06 214, 3.55

$V_{gc} = -1.0$
52, 0.11 76, 0.28 108, 0.59 129, 0.86 155, 1.24 178, 1.64 211, 2.32 246, 3.12

$V_{gc} = -1.5$
97, 0.11 137, 0.40 169, 0.76 193, 1.13 226, 1.63 277, 2.66

$V_{gc} = -2.0$
142, 0.12 171, 0.30 206, 0.62 243, 1.10 281, 1.73 311, 2.33

$V_{gc} = -2.5$
183, 0.12 208, 0.24 237, 0.46 265, 0.74 294, 1.13 321, 1.56 342, 1.96

$V_{gc} = -3.0$
225, 0.12 248, 0.20 280, 0.40 313, 0.72 339, 1.05 372, 1.57

$V_{gc} = -3.5$
284, 0.12 308, 0.27 334, 0.46 370, 0.81 397, 1.17

$V_{gc} = -4.0$
319, 0.12 342, 0.22 372, 0.38 402, 0.65 425, 0.90

$V_{gc} = -4.5$
363, 0.13 388, 0.22 413, 0.36 446, 0.62

$V_{gc} = -5.0$
405, 0.12 421, 0.17 442, 0.26 463, 0.40

first derivative in the area where typical load lines are known to live. This is of concern, since the poles and zeros of coupled analog gain circuits depend primarily on the dynamic properties of the devices and circuit elements used. In fact, the dynamic properties of all devices are related and defined directly in terms of the slope profile of their transfer characteristics.

Furthermore, the amount of simulation-error sensitivity in this area would be higher in designs that avoid using any kind of global feedback. For example, models with poor absolute or slope matching cannot be relied upon for exacting tasks such as the design of single-ended, open-loop, RIAA-compensated phono preamp circuits. It follows that such a divergence in curvature characteristics is most likely to produce many numerical errors in circuit simulations in general.

New Model for VT Triodes

To solve such problems, I propose a new model based on a complete definition of two-port networks and including a freer algebraic expansion of the transfer-characteristic functions. Moreover, the model derives directly from device data, which allows for full custom modeling as long as the data is available in tabular form.

The functional aspect of the model is covered by the principle that a two-port network is deemed "specified" over a given input-voltage variable range if you can determine port currents in terms of port voltages over that range. I will demonstrate the model for a 12AX7 triode, though you can easily apply it to any other type of triode (*Fig. 2*).

For a triode, a two-port representation means specifying I_p and I_g current functions in terms of both V_{gc} and V_{pc} voltage differences. This two-port formulation must hold over the device's intended range of application, one that is as wide as possible so that you can use the model all the way into the hard-overload applications to study distortion characteristics.

Also, the model fit should be accurate over the main regions of application in absolute as well as incremental terms; that is, static and dynamic levels should fit well against corresponding data functions. These three conditions are sufficient to provide a functionally correct and accurate device model for triodes (*Fig. 3*).

RCA 12AX7 Reference Data

In this work I use tube data provided by the RCA catalog[7] for modeling the 12AX7 triode. This choice is made out of convenience because it lends itself well to a representative formulation of the method. It would be ideal to test, average, and model individual brands and types of tubes, as is the case with transistors. When tube data becomes available from the current tube industry—



FIGURE 4: Second level interpolation of first level coefficients against input voltage V_{gc}.

ABOUT THE AUTHOR

J.C. Maillet studied ergodic theory with Jal Choksi in 1986 and current-mode signal processing and current-feedback op-amp design with Gordon Roberts in 1991, both at McGill University. He later went on to design ultra-low-jitter high-speed analog PLLs for PMC-Sierra's S/UNI-Lite Internet transceiver chip. Currently living in his Dodge van exploring the remote corners of British Columbia, he does tube-amp tech work at Arbutus Music and Galahad Studios in Nanaimo, B.C.

TABLE 2

12AX7 PLATE-CURRENT DATA VS. INPUT VOLTAGE, 1V TO –5V

FOURTH-ORDER INTERPOLATION

CP1P0 =	–2.2579e–03	4.6069e–02	–2.0346e–01	4.9278e–01	–6.8805e+00
CP0P5 =	7.0981e–03	–5.7696e–02	1.0537e–01	7.7425e–01	–9.1508e+00
CP0P0 =	8.8347e–03	–7.2027e–02	1.2145e–01	9.5389e–01	–1.0093e+01
CM0P5 =	2.5971e–03	–2.9848e–02	1.6737e–01	7.4775e–01	–1.2005e+01
CM1P0 =	–6.9709e–03	4.0845e–02	2.1593e–01	3.4068e–01	–1.4616e+01
CM1P5 =	–1.7082e–02	8.2555e–02	5.7013e–01	–1.4974e–01	–2.0729e+01
CM2P0 =	–2.4575e–02	1.3531e–01	6.8172e–01	–1.1425e–01	–2.6793e+01
CM2P5 =	–2.6901e–02	1.5198e–01	7.7130e–01	1.3492e–01	–3.2297e+01
CM3P0 =	–1.7776e–02	4.9488e–02	1.0619e+00	1.3333e+00	–3.9986e+01
CM3P5 =	–8.7813e–03	–3.0020e–02	1.1642e+00	2.7989e+00	–4.7449e+01
CM4P0 =	4.6364e–03	–1.6978e–01	1.2942e+00	5.5445e+00	–5.6533e+01
CM4P5 =	1.6994e–02	–2.9338e–01	1.3500e+00	8.3722e+00	–6.5618e+01
CM5P0 =	4.0726e–02	–5.5315e–01	1.8574e+00	1.1129e+01	–7.6001e+01

TABLE 3

COEFFICIENTS VS. ORDER OF DATA

SEVENTH-ORDER INTERPOLATION

K0 =	3.3101e–3	6.4276e–2	4.7240e–1	1.5643e+0	1.8661e+0	–2.8135e+0	1.9145e+0	–9.9158e+0
K1 =	1.3632e–3	1.8921e–2	1.0492e–1	2.6213e–1	–4.8578e–2	–8.3349e–1	3.2558e–2	9.5428e–1
K2 =	–1.4882e–3	–2.0981e–2	–1.0960e–1	–2.4952e–1	–1.7040e–1	2.2391e–1	2.5192e–2	9.5766e–02
K3 =	3.4761e–4	4.2512e–3	1.7002e–2	2.4265e–2	3.1025e–2	7.5560e–2	–3.9657e–2	–6.6107e–2
K4 =	–2.4359e–5	–2.4727e–4	–5.6853e–4	5.2888e–4	–1.9288e–3	–1.3258e–2	4.7989e–3	8.4148e–3

TABLE 4

SUBCIRCUIT MODEL FOR "EXTENDED" 12AX7 TRIODE

```
************************ p g c
************************ .subckt RCA12AX7 1 2 3 ************************
eGlogVpc 20 0 value={log(v(1,3))}
rGlogVpc 20 0 1
eG0 10 0 poly(1) <2,3>  -3.7694e+00  1.9947e+00  5.9432e-02
eG1 11 0 poly(1) <2,3>  -3.2024e-02  -4.1443e-02  -4.8236e-03
eG2 12 0 poly(1) <2,3>  1.9127e-02  -1.2189e-02  -1.5526e-03
eG3 13 0 poly(1) <2,3>  -1.1354e-02  4.9339e-03  6.1016e-04
rG0 10 0 1
rG1 11 0 1
rG2 12 0 1
rG3 13 0 1
gG  2 3 value={(exp(v(10)+ v(20)*(v(11)+ v(20)*(v(12)+ v(20)* v(13)))))/170}
*
eP0 110 0 poly(1) <2,3>  -9.9158e+0 1.9145e+0 -2.8135e+0 1.8661e+0 + 1.5643e+0
+                       4.7240e-1 6.4276e-2 3.3101e-3
eP1 111 0 poly(1) <2,3>  9.5428e-1 3.2558e-2 -8.3349e-1 -4.8578e-2 + 2.6213e-1
+                       1.0492e-1 1.8921e-2 1.3632e-3
eP2 112 0 poly(1) <2,3>  9.5766e-2 2.5192e-2 2.2391e-1 -1.7040e-1 + -2.4952e-1
+                       -1.0960e-1 -2.0981e-2 -1.4882e-3
eP3 113 0 poly(1) <2,3>  -6.6107e-2 -3.9657e-2 7.5560e-2 3.1025e-2 + 2.4265e-2
+                       1.7002e-2 4.2512e-3 3.4761e-4
eP4 114 0 poly(1) <2,3>  8.4148e-3 4.7989e-3 -1.3258e-2 -1.9288e-3 + 5.2888e-4
+                       -5.6853e-4 -2.4727e-4 -2.4359e-5
rP0 110 0 1
rP1 111 0 1
rP2 112 0 1
rP3 113 0 1
rP4 114 0 1
gP  1 3 value={(exp(v(110)+v(20)*(v(111)+v(20)*(v(112)+v(20)*(v(113)+v(20)*
+                       v(114)))))))}
Cgc 2 3 1.8p
Cgp 2 1 1.7p
Cpc 1 3 1.9p
.ends
```



FIGURE 5: New-model transfer output against RCA data, with additional positive-input voltage curves.

most probably extracted with better techniques[8] than those used in the late 50s—the algorithms provided here could be used to develop SPICE models.

Because many of us are especially curious about tube behavior in the overload region, the curves RCA supplies need extension into the positive range of input voltages. Though the grid-shift technique is prone to error in extracting complete sets of transfer curves, I used it on a GE 12AX7 to mark out two extra I_p (V_{pc}) curves RCA does not provide. I used a GE tube simply because I didn't have an RCA 12AX7 lying around at the time. I'm sure the values are close enough for demonstration purposes.

The model possesses a voltage range of –5 to +1V on the input port (V_{gc}), but I must make clear that outside this range the model has little guarantee of reliability. I vectorized the data used in generating the model from a random choice of point pairs lying on the curves. You should note that the choice of data points will influence the outcome of the final model.

Because grid-current levels are known to vary by as much as two orders of magnitude from tube to tube within the same manufacturing lot, there is no reason to be too fussy with this side of the data, as long as it corresponds moderately well to that of a real device. I use 6v6 pentode tube grid-current specs from the RCA data book and scale them down roughly to values I've measured with 12AX7 tubes on the bench. Curve profiles appear similar enough in my tests so that I use them interchangeably, although in official models, plate and grid data would be properly measured from the same tube.

Polynomial Interpolation Techniques

Both the grid- and plate-current functions of vacuum-tube triodes exhibit transfer characteristics that make them ideal for polynomial interpolation; that is, both current-function dependencies to output-port voltage (V_{pc}) are smooth and knee-less, unlike transistor- and pentode-device curves, and their evolution from one constant input-voltage curve to the next is very gradual and smooth in dependence to input-port voltage (V_{gc}). You can exploit these two simple traits to derive a model based on a polynomial interpolation of the vectorized data set.

The advantage of using polynomial functions lies in their ability and freedom to approximate all continuous functions accurately, whereas the range covered by 3/2-power-based and many other fixed-order functions is greatly restricted in comparison, thereby rendering incomplete their corresponding "approximation space." Because polynomial interpolation involves weaving through point pairs, you must convert the data to log scale in order to avoid negative current values when approaching the x-axis, as well as to help achieve an overall reduction of the effects of weaving.

With or without this logarithmic-scale conversion, the basic idea behind the use of polynomial interpolation remains the same. That is, each curve making up the transfer set has a set of point pairs through which you can extract a set of polynomial coefficients. I use Matlab for this purpose.[9] This resulting set of coefficients depends directly on the original choice of point pairs taken from the curves (*Table 1*). To every I_p (V_{pc}) or I_g (V_{pc}) curve there corresponds a group of coefficients that you can list in tabular form against the input voltage V_{gc} (*Table 2*).

Interpolating each data vector to the same order each time gives a very specific meaning to the column values of *Table 2;* namely, each column represents a functional relationship between a par-

to page 34

from page 31

ticular coefficient order and the input-voltage variable V_{gc}. Because the columns are also in vectorized form, you can interpolate them polynomially in the same way (possibly to a different order) to yield corresponding coefficient sets (*Table 3*).

You can use this last set of coefficients to generate not only the previous set–by expanding the polynomial expression at the values of input voltage (V_{gc}) corresponding to the data–but also those coefficients corresponding to voltages lying between voltage values chosen to represent the data. Provided you choose the order properly in the second interpolation, you can produce a fill-in that follows along with the general progression of the coefficient values themselves. The ability of each function to interpolate the first-level coefficients in a well-behaved manner is depicted in *Fig. 4*.

General Form of Equation for Current Function

If the interpolation at this second level weaves properly through the coefficient values of the first-level interpolation, then you can certainly recover the original data from the last set of coefficients alone through reverse steps of expansion. More importantly, this means that you can also simulate intermediate values with a high degree of reliability (which you need to test when checking the models). With the log conversion and subsequent expansion taken into account, you can write a general form for the converted current function in the following computationally economical polynomial form:

$$\text{for } x = p \text{ or } g,\ Ix(\log(V_{pc}),V_{gc}) = \mathrm{Exp}(\log(Ix(\log(V_{pc}),V_{gc}))) = \mathrm{Exp}(Ko(V_{gc}) + K1(V_{gc})(\log(V_{pc})) + K2(V_{gc})(\log(V_{pc})) + K3(V_{gc})(\log(V_{pc})) + K4(V_{gc})\log(V_{pc})))) \quad [1]$$

One advantage provided by the log-scale conversion of the data is a reduction in transfer-set curvature, thus allowing increased accuracy at lower interpolation orders. On the other hand, this prevents the range from reaching down to the limit V_{pc} = 0V. So you must establish an arbitrary low-voltage minimum (0.1V) for the second variable's domain and set the data by hand in the area of very low current and voltage (near the origin) as a means of guiding the polynomials appropriately through the more important and relevant operating-voltage regions. This is the model's only disadvantage, but it can cause much trouble whenever the simulator rolls back the power-supply voltage at the beginning of a simulation.

For this reason, you must set initial values properly in many simulations if failure to converge is not to occur. The reason? The model now contains two active ports, each with a mind of its own. If one port sends its current function in the wrong direction too fast, before the other port can make up its mind, then circuit voltages will arise so that the model cannot recover itself. Furthermore, convergence becomes impossible if the terminal-voltage differences are sent to a nonmonotonic part of one of the terminal-current functions where folds or spatial condensation can produce a trap for the searching algorithms. To date, there is no general solution to this problem except for that involving near-origin data manipulation, as some of my own experiments have suggested.

Since the polynomial interpolation into coefficients and subsequent expansions both form a chain of continuously monotonic and reversible operations, the composition of all these operations itself forms a continuous operation on the original data. This resulting composition approximates the identity function f(x) = x, provided you perform each step with enough resolution power. Also, if the coefficient functions K_i (V_{gc}) are well behaved in the sense that the polynomial function interpolating them doesn't go wild between target values, then you can expect the identity function to carry over in between the original data curves themselves, and so provide a good fit there as well.

PSpice Subcircuit Model

In the PSpice[10] subcircuit for the full model (*Table 4*) are the grid- and plate-current functions that provide both the functional mechanisms and transfer characteristics of the device according to the general two-port and functional formulation described above. Notice the repeated use of the polynomial functions, which are computationally efficient expressions in SPICE, as well as the sporadic use of the less efficient logarithmic and exponential functions.

The subcircuit uses controlled-voltage sources to implement the polynomial functions and coefficients, while a transconductance function converts the last step to port current. Dummy termination resistors on the inner polynomial controlled-source circuits avoid simulator inductor loops. Also, static capacitance values are provided as crude representatives of interterminal parasitics, although you could use interpolation techniques similar to those utilized for terminal current to model them if the need arises.

It is worth mentioning that the freely



FIGURE 6a: Grid-current response of new model; linear scale.



FIGURE 6b: Grid-current response of new model; log scale.



FIGURE 6c: Plate-current response of new model showing problem area at low-voltage/ low-current values; log scale.

downloadable PSpice 5.0/6.0 demo package does not penalize controlled-source calls as much as, say, a diode subcircuit call. This means you can simulate larger circuits in such restricted platforms with this approach. In fact, by using these models, I have simulated entire Fender amplifiers under all types of analysis, while yielding very acceptable process times on my modest Macintosh Quadra 630.

Superimposing PSpice-modeled output on the "extended" RCA data used to derive the model (*Fig. 5*) reveals that the fit is fairly accurate in both absolute and first-order terms. This degree of accuracy allows you to closely determine some important triode circuit characteristics through simulation.

Figures 6a and *6b* show two grid-current plots so you can become familiar with their profiles. The grid-current function is monotonic down to −5V, which is as important for proper behavior of the model as it is for the plate-current function. *Fig. 6c* shows the log-scale version of the plate-current function with the low-power problem area that currently needs improving.

Large-Signal SPICE Simulations

The model presented here was derived primarily to determine accurately the intrinsic nonlinear drive and load properties of triode tube circuits, and also for examining the effects of grid-current-loading and device-transfer nonlinearities in large-signal cascaded-triode gain-stage circuits. This is where specific types of nonlinear device characteristics meet to produce some of the circuit behavior responsible for providing tube circuits with their signature feel.

You can obtain the transfer characteristics of a common-cathode gain stage (*Fig. 7*) through a swept DC analysis, while large-signal simulations of cascaded circuits are typically simulated using transient analysis to produce results as they would appear on an oscilloscope on the bench in order to capture circuit time dependencies.



FIGURE 7: Typical common-cathode gain-stage biasing.

Common-Cathode Transfer

As an example, you can simulate the primary drive characteristics of a common 100k/1.5k @ 300v gain stage employing a 12AX7 tube to extract the exact wide-range circuit-transfer function. Then you can use this function later to generate Fourier coefficient evolution against source amplitude to produce accurate harmonic-distortion characteristics if you consider them of value.

A DC sweep will generate the circuit's important voltage and current variable transfer functions against the circuit's input voltage—in this case ranging from −6V to +6V. Notice how the device's input-port voltage (V_{gc}) varies against circuit-input voltage (V_g), and note as well the exponential rise in grid current for positive input voltage (*Fig. 8*).

No longer is the input port modeled as a misrepresentative sensing circuit. Grid and plate current are added together to form the cathode current, which in turn produces dynamic-feedback mechanisms in the gain stage that are partly responsible for compression and harmonic-distortion effects. These effects are well-known to increase with signal amplitude, although no one has previously observed or described them accurately due to the difficulties in testing for them.

Cascaded Common-Cathode Circuits

By simulating the front end of a cascaded triode circuit (*Table 5*), as seen in many amplifier circuits, you can highlight a property of triode tube circuits that is particularly relevant to electric guitarists. Of particular interest here is determining—by simulating the cascading of two gain stages as they would appear in standard amp circuits—how grid-current response changes against key volume-control settings. This change, which in turn reflects a dynamic-response-factor change, seems to occur in practice when the volume-control pot is somewhere between full on and 90% rotation.

The test involves applying a 1kHz sine-wave current to the grid of the first stage, which is similar to how a guitar or phono pickup would deliver energy to the front end of the circuit. Here the intention is to observe the waveform at the first plate as a function of second-stage loading as an indication of large-signal distortion (*Fig. 9*). By setting the input source to a sufficiently large amplitude, you overload the second stage so that the second grid circuit sources appreciable amounts of current relative to the plate circuit of the first stage.

In the case of 100% rotation, the volume pot simply acts as a 1MΩ load to ground, with no series resistance between the plate and grid of the two stages. In the 90% rotation, a 10% reduction in grid-voltage amplitude occurs—because of the pot acting as a voltage divider—while introducing a 100kΩ resistor into the signal path. This series resistive component between successive plate and grid terminals is known to attenuate grid-current production in a subsequent gain stage. For the same reason, grid-blocking resistors are almost always found on power-tube grid terminals to



FIGURE 8: Transfer variables of common-cathode gain stage against swept input source.



FIGURE 9: Typical cascaded gain stages with current source and volume control.

protect the tube from input-circuit power overload.

Though a 10% voltage-signal loss has occurred, you see a 50% reduction in grid current, which is characteristic of nonlinear circuit behavior. This relative difference suggests a disproportionate increase in input-circuit loading on the output circuit of the previous gain stage when you turn the pot full on.

This test lays a foundation for explaining why Fender Blackface amplifiers have a smoother, less distortive response compared to the older Fender Tweed amps. This is so because of circuit compression and distortion sensitivity to series interstage resistance levels. For example, any Tweed Champ turned on full exhibits a total lack of series resistance throughout the circuit, and so exhibits undeniable amounts of dynamic compression and distortion sweetness. I believe this circuit characteristic is also responsible for the glassy tonal quality that clean-topology Tweed amps possess at lower operating levels.

Further step-transient simulations can show that grid current plays a crucial role in establishing compressive effects (*Fig. 10*). In fact, other simulations have shown that plate-transfer nonlinearities together with grid-current characteristics work vectorially in the same direction towards establishing large-signal compression effects; that is, the direction of the transfer curvature of the plate-current function as well as the grid-current response to positive input-voltage swings both act to reduce gain in the stage.

Conclusion

The work presented here offers a radical departure from previous vacuum-tube modeling techniques in that it is both more algebraically "general" in nature and based directly on tube data. Special mention should be made of the fact that the equations used to perform the modeling are of a most simple nature and can be understood by anyone willing to spend a little time with the inner algorithms. Although the model has been



FIGURE 10: Plate-voltage and grid-current compression occurring at interface of cascaded gain stages.

shown to work, a few areas need improvement before this technique can produce official models that also have good converging properties.

First, a refinement in the choice of discrete data points needs to be established in a nonrandom fashion; second, a method needs to be developed for placing low-voltage/low-current data values for influencing best fit of the model in the more important parts of its terminal-voltage domain—that is, below the maximum power-dissipation curve, somewhat away from the origin, the y-axis, the high plate-voltage area.

This is not such a simple task to perform, especially when you are randomly choosing the data points and algorithm parameters, as I have done so far. Because of this, the method is currently in a hit-or-miss state. Further refinements will probably be discovered by employing standard Monte Carlo techniques.

It is worth noting that the new modeling technique has also been applied to the RCA 6.6V pentode-tube plate data (screen tied to 250V). The strong nonlinearity near the x-axis proves to be a tough challenge, and the transition to lower negative values is difficult to keep near zero in a monotonic manner, though some similarity has been achieved over a limited range of terminal voltages.

Because of the knee in the response of pentodes, and the sharp cutoff near the x-axis, the algorithms find themselves trying much harder to get into shape for the match. It might even be possible to use the types of nonlinear "kneed" functions existing in current pentode models as molds upon which to add or multiply a fitting function based on the polynomial form presented here.

At a macrocircuit level, you could achieve polynomial two-port modeling of entire or even cascaded gain stages, with proper AC and DC large-signal drive and loading characteristics as depicted in *Fig. 8*—and much simpler to interpolate. This you can now do through circuit simulation from accurate models instead of testing. This could possibly pave the way for a bipolar transistor implementation—via the translinear circuit principle[11] and a voltage-scale network transformation—of semiconductor circuits that realistically exhibit the feel and overload characteristics of tube circuits.

The author would like to thank Dr. J. Choksi of the McGill University mathematics department, Dr. H.C. Lee and Dr. G.W. Roberts of the same university's electrical engineering department, and N.S.E.R.C. and MicroNet of Canada for past support. ◆

SOURCES

You can download Matlab and PSpice source files used in this work at http://www.lynx.bc.ca/~jc/ or by E-mail from the author directly at jc@lynx.bc.ca.

REFERENCES

1. Dr. W. Marshall Leach, Jr., "SPICE Models for Vacuum Tube Amplifiers," JAES, Vol. 43, No. 3 (March 1995) pp. 117–126.

2. S. Reynolds, "Vacuum-Tube Models for SPICE Simulations," *GA* 4/93, pp. 17–23.

3. N. Koren, "Improved VT Models for SPICE Simulation," *GA* 5/96, pp. 18–27, 69.

4. C. Rydel, "Tube Modeling," *GA* (Letters) 6/96, pp. 69–71.

5. E. Pritchard, "Tube-Model Critique", *GA* (Letters) 1/96, pp. 54–57.

6. J. Tolonen, "High-Performance Tubed DAC Output Stage," *GA* 2/97, pp. 22–31.

7. *RCA Tube Manual*, 1959.

8. Private conversation with tube-audio consultant Jim Atkins (Hart Magnetics) regarding modern testing of tubes for data.

9. Matlab, Student Edition, The MathWorks, Inc. 1992.

10. PSpice 6.0, MicroSim Corporation, 1994.

11. E. Seevinck, *Analysis and Synthesis of Translinear Integrated Circuits*, Elsevier Science Ltd., Jan. 1988.

TABLE 5

PSPICE DECK FOR TWO-STAGE CASCADED CIRCUIT

```
Cascaded gain stages with RCA 12AX7 SPICE model
*
Vsupply 99 0 300
* input pickup represented as ideal sinusoidal current source
Iin 0 1 sin(0 1e-6 1kHz 0)
*input resistor
       *low input 2
       Rg1 1 2 68k
       *high input 1
       *Rg1 1 2 34k
*first gain stage
       Rg2 2 0 1Meg
       Rc1 3 0 1.5k
       Rp1 99 4 100k
       x1 4 2 3 RCA12AX7
       Cs1 4 5 .02u
* pot at 100% rotation
       Rvol1 5 6 1
       Rvol2 6 0 1Meg
* pot at 90% rotation
       *Rvol1 5 6 100k
       *Rvol2 6 0 900k
*second gain stage
      *zero volt source used as OW current
      Vig2 6 96 0
       x2 8 96 7 RCA12AX7
       Rp2 99 8 100k
       Rc2 7 0 1.5k
*** insert 12AX7 tube model here
*initial conditions
.ic v(2)=-0.570591 v(3)=1.4180 v(4)=230.526 v(5)=-0.570591
+   v(6)=-0.570591 v(7)=1.4180 v(8)=230.526
*transient and output processor calls in PSpice
.tran 1e-6 1e-3 0 1e-6
.probe
.end
```